
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc92828192]Corrections required
0.1 [bookmark: _Toc92828193]Correction to function f_TC_11_1_5_NR5GC_IMS1
	Function name
	f_TC_11_1_5_NR5GC_IMS1

	Reason for change
	The changes mentioned in this document are needed after implementing changes mentioned in R5s211706 (Except Change 8).
In the TTCN implementation after implementing the R5s211706, there are still some coordination messages which are needed/not needed.
This needs to be fixed.

	Summary of change
	Removed a coordinatin message from f_TC_11_1_5_NR5GC_IMS1 which was waiting for a trigger from Eutra, and added a coordination message to Eutra which was needed after reception of Attach Complete.

	TTCN module
	EPS_Fallback_NR5GC_IMS.ttcn

	MCC160 Comment
	Not needed



Before Change:
	function f_TC_11_1_5_NR5GC_IMS1()runs on IMS_PTC
  {
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_CoordMsg;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) MessageHeader v_MessageHeader_Response183;
    var default v_MyDefaultVar := null;
    
    fl_IMSFallback_PreambleRegistrations();

    // @siclog "Step 11" siclog@
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_ReceiveINVITE_Common(-, -);
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s211660 sic@
    // Wait for indication RAT has changed to EUTRA
    OtherIPCAN.receive(cmr_IPCAN_IMS_Response) -> value v_IMS_IPCAN_CoordMsg;
    f_IMS_PTC_SetRanType(v_IMS_IPCAN_CoordMsg.IPCAN_IMS_Response.IpcanInfo.RanType);
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // @sic R5s211660 sic@

    //@siclog "Step 28 - 29" siclog@

    v_MyDefaultVar := activate (a_C46_ParallelBehavior ());
    //Steps 28 - 29 Map to 36.508 Table 4.5A.6.3-1 Steps 9a1- 11a1 which maps to 34.229-1 C.21 Steps 3 -4
    f_IMS_MOCallSetup_Send100Trying(v_InviteRequestWithSdp);

    v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequestWithSdp.Invite, px_IMS_CalleeContactUri, omit, -, NORMAL_CALL);
    f_IMS_MOCallSetup_AnnexC21_Step4(v_InviteRequestWithSdp, v_MessageHeader_Response183); // @sic R5s211660 sic@
    deactivate (v_MyDefaultVar);
    // Trigger EUTRA for Dedicated Bearer Setup
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    //Carry on with the C.21 Steps  5-13
    f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, valueof(v_MessageHeader_Response183));
       
    // @siclog "Step 36" siclog@
    f_Delay(1.0);
    
    // @siclog "Step 37" siclog@
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5s210662 sic@
  }



After Change:
	function f_TC_11_1_5_NR5GC_IMS1()runs on IMS_PTC
  {
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_CoordMsg;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) MessageHeader v_MessageHeader_Response183;
    var default v_MyDefaultVar := null;
    
    fl_IMSFallback_PreambleRegistrations();

    // @siclog "Step 11" siclog@
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_ReceiveINVITE_Common(-, -);
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); // @sic R5s211660 sic@
    // Wait for indication RAT has changed to EUTRA
    OtherIPCAN.receive(cmr_IPCAN_IMS_Response) -> value v_IMS_IPCAN_CoordMsg;
    f_IMS_PTC_SetRanType(v_IMS_IPCAN_CoordMsg.IPCAN_IMS_Response.IpcanInfo.RanType);
    
    //WA#11_1_5 COMMENTED OUT This is not needed f_IMS_IPCAN_WaitForTrigger(OtherIPCAN); // @sic R5s211660 sic@
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);   //WA#11_1_5 : This is needed as Eutra waits for this trigger after Attah complete    

    //@siclog "Step 28 - 29" siclog@

    v_MyDefaultVar := activate (a_C46_ParallelBehavior ());
    //Steps 28 - 29 Map to 36.508 Table 4.5A.6.3-1 Steps 9a1- 11a1 which maps to 34.229-1 C.21 Steps 3 -4
    f_IMS_MOCallSetup_Send100Trying(v_InviteRequestWithSdp);

    v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequestWithSdp.Invite, px_IMS_CalleeContactUri, omit, -, NORMAL_CALL);
    f_IMS_MOCallSetup_AnnexC21_Step4(v_InviteRequestWithSdp, v_MessageHeader_Response183); // @sic R5s211660 sic@
    deactivate (v_MyDefaultVar);
    // Trigger EUTRA for Dedicated Bearer Setup
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    //Carry on with the C.21 Steps  5-13
    f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, valueof(v_MessageHeader_Response183));
       
    // @siclog "Step 36" siclog@
    f_Delay(1.0);
    
    // @siclog "Step 37" siclog@
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite);
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5s210662 sic@
  }


0.2 [bookmark: _Toc92828194]Correction to function fl_TC_11_1_5_NR5GC_EUTRA_TestBody
	Function name
	fl_TC_11_1_5_NR5GC_EUTRA_TestBody

	Reason for change
	The changes mentioned in this document are needed after implementing changes mentioned in R5s211706 (Except Change 8).
In the TTCN implementation after implementing the R5s211706, there are still some coordination messages which are needed/not needed.
This needs to be fixed.

	Summary of change
	Removed a coordination message waiting for a trigger from IMS PTC in function fl_TC_11_1_5_NR5GC_EUTRA_TestBody which was not required and was stalling the establishment of Dedicated EPS bearer for speech.

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	Alternative solution implemented: don’t wait for IMS signalling atter sending the Attach Complete message



Before Change:
	function fl_TC_11_1_5_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var EPS_BearerIdentity v_EpsBearerId; // @sic R5s211660 sic@
    var SRB_COMMON_IND v_ReceivedAsp; // @sic R5s211660 sic@
    timer t_Wait := 5.0; // @sic R5s211660 sic@

    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell3})); // @sic R5s211660 sic@
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 9A" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE_FR2); // @sic R5s211660 sic@
    f_EUTRA38_SetCellPower(eutra_Cell3, tsc_CellRS_EPRE_73, tsc_CellRS_EPRE_73);
    // @siclog "Step 14 - 27" siclog@
    fl_TAUReject_AttachWithoutN26 ();

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // @sic R5s211660 sic@
    t_Wait.start; // @sic R5s211660 sic@
    alt { // @sic R5s211660 sic@
      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover, ?)))) -> value v_ReceivedAsp
        {
          v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 (eutra_Cell1, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST);
          a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(eutra_Cell1);
          a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(eutra_Cell1, v_EpsBearerId);
        }
      [] t_Wait.timeout
        {
        }
    }

    // @siclog "Step 30 -32" siclog@
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }



After Change:
	function fl_TC_11_1_5_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var EPS_BearerIdentity v_EpsBearerId; // @sic R5s211660 sic@
    var SRB_COMMON_IND v_ReceivedAsp; // @sic R5s211660 sic@
    timer t_Wait := 5.0; // @sic R5s211660 sic@

    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell3})); // @sic R5s211660 sic@
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 9A" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE_FR2); // @sic R5s211660 sic@
    f_EUTRA38_SetCellPower(eutra_Cell3, tsc_CellRS_EPRE_73, tsc_CellRS_EPRE_73);
    // @siclog "Step 14 - 27" siclog@
    fl_TAUReject_AttachWithoutN26 ();

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // @sic R5s211660 sic@
    t_Wait.start; // @sic R5s211660 sic@
    alt { // @sic R5s211660 sic@
      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover, ?)))) -> value v_ReceivedAsp
        {
          t_Wait.stop;
          v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 (eutra_Cell1, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST);
          a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(eutra_Cell1);
          a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(eutra_Cell1, v_EpsBearerId);
        }
      [] t_Wait.timeout
        {
        }
    }

    // @siclog "Step 30 -32" siclog@
    //WA#11_1_5 COMMENTED OUT f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }


MCC160 Implementation:
  function fl_TC_11_1_5_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var EPS_BearerIdentity v_EpsBearerId; // @sic R5s211660 sic@
    var SRB_COMMON_IND v_ReceivedAsp; // @sic R5s211660 sic@
    timer t_Wait := 5.0; // @sic R5s211660 sic@

    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell3})); // @sic R5s211660 sic@
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 9A" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE, tsc_ServingCellRS_EPRE_FR2); // @sic R5s211660 sic@
    f_EUTRA38_SetCellPower(eutra_Cell3, tsc_CellRS_EPRE_73, tsc_CellRS_EPRE_73);
    // @siclog "Step 14 - 27" siclog@
    fl_TAUReject_AttachWithoutN26 ();

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // @sic R5s211660 sic@
    t_Wait.start; // @sic R5s211660 sic@
    alt { // @sic R5s211660 sic@
      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover, ?)))) -> value v_ReceivedAsp
        {
          t_Wait.stop;
          v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 (eutra_Cell1, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST);
          a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(eutra_Cell1);
          a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(eutra_Cell1, v_EpsBearerId);
        }
      [] t_Wait.timeout
        {
        }
    }

    // @siclog "Step 30 -32" siclog@
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }
  
  function fl_TAUReject_AttachWithoutN26 () runs on EUTRA_Multi5GS_PTC
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var template (present) NAS_MSG_Indication_Type v_NAS_Expected;
    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;
    var template (value) MobileIdentity v_MappedGUTI;
    var GutiParameters_Type v_GutiParams;
    var template AdditionalUpdateType v_AdditionalUpdateType;
    var NAS_PlmnId v_PLMN;
    var TrackingAreaCode v_Tac;
    var template TrackingAreaId v_TAI:= omit;
    var template (present) NAS_KsiValue v_KsiValue;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    var template (omit) integer v_SessionId;
        
    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);
    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);
    v_PduSessionInfoList := f_EUTRA38_WaitForPDUSessionInfoList(eutra_Cell1); // @sic R5s210149 sic@
    
    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

    v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);
    v_Tac := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);
    v_TAI := cr_TAI(v_PLMN, bit2oct(v_Tac));
   
    // @siclog "Step 7 - 8" siclog@
    v_NAS_Expected := (cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                          cr_TAU_Request_N1ToS1(f_GetEPSTAUType(NORMAL),
                                                                v_KsiValue,
                                                                cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                v_AdditionalUpdateType,
                                                                v_MappedGUTI,
                                                                v_TAI,
                                                                tsc_EpsUpdate_Active,
                                                                -, -,
                                                                omit,
                                                                cr_UeRadioCap ifpresent)),
                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                         cdr_ATTACH_REQUEST_N1ToS1(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_KsiValue,
                                                                                   f_GutiParameters2MobileIdentity(omit, v_GutiParams), // @sic R5s211706 sic@
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType())));
    
    // @siclog "Step 8a1/8b1" siclog@
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_Cell1,
                                               tsc_RRC_TI_Def,
                                               ?,
                                               v_NAS_Expected);

    if (ischosen (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST)) {
        // @siclog "Step 8b2" siclog@
        SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,
                                                  cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_EmmCause_UEIdentityNotDerived))));

        // @siclog "Step 8b3" siclog@
        SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                   cdr_ATTACH_REQUEST_IMSIonly(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType()))))
                            -> value v_ReceivedAsp;
        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
    }

    // @siclog "Step 9 - 20" siclog@
    v_SessionId := f_ProtocolConfigOptions_GetSessionId(v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions);
    if (not isvalue (v_SessionId)) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Id not present in PCO");
    }

    f_EUTRA38_MultiPDN_InitialRegistration_Step5_16 (eutra_Cell1, TESTMode_OFF, v_NasInd, -, v_PduSessionInfoList, true); // @sic R5s210149 sic@

  }

  function f_EUTRA38_MultiPDN_InitialRegistration_Step5_16(EUTRA_CellId_Type p_CellId,
                                              NR_TESTMODE_STATE_Type p_State,
                                              NAS_MSG_Indication_Type p_NAS_Ind,
                                              template (omit) PLMN_List p_EquivalentPlmnList := omit,
                                              template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduSessionInfoList := omit,
                                              boolean p_FallbackSignalling := false) runs on EUTRA_Multi5GS_PTC
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var template (omit) AdditionalUpdateType v_AdditionalUpdateType;
    var NAS_UL_Message_Type v_PdnConnectivityRequest;
    var ProcedureTransactionIdentifier v_EPS_TI;
    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;  // may be omit
    var template (omit) ExtdProtocolConfigOptions v_ExtdPcoFromUE := omit;
    var EPS_BearerIdentity v_EpsBearerId;
    var boolean v_IPv4AllocationViaNasFlag;
    var APNandPCOs_Type v_APNandPCOs;
    var NasCount_Type v_NasCountUL;
    var NR_Configuration_Type v_NR_ConfigType := f_EUTRA38_NR_Config_Get();
    var DelayForUserPlaneSignalling_Type v_WaitForUserplaneSignalling := noDelay; // Default: no IP signalling possible
    var template (omit) UERadioCapId v_UERadioCapId:= omit;

    if (not ispresent (p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.ueAddSecurityCap)){
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "ueAddSecurityCap should be present");
    }
    select (v_NR_ConfigType) {
      case (EN_DC) {
        if (not match (p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.ueNetworkCapability, cdr_EUTRA_UENetworkCap(omit))){
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "ueNetworkCapability does not indicate support for ENDC");
        }
      }
      case else { // other options FFS
        if (not match (p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.ueNetworkCapability, cdr_NR5GC_UENetworkCap(omit))){
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "ueNetworkCapability does not indicate support for EUTRA38");
        }
      }
    }
    
    v_AdditionalUpdateType := p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType;
    v_PdnConnectivityRequest := p_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;
    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    v_PcoFromUE := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;
    v_ExtdPcoFromUE := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;
    
    // Steps 5-10b6
    v_ReceivedAsp := f_EUTRA_MultiPDN_InitialRegistration_Step5_8 (p_CellId, p_NAS_Ind);  // @sic R5-216280 sic@
    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;
    v_APNandPCOs := f_EUTRA38_MultiPDN_InitialRegistration_Step9_13 (p_CellId, p_State, p_NAS_Ind, v_NasCountUL);
    
    if (ispresent(v_APNandPCOs.protocolConfigurationOptions)) {
      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;
    } else if (ispresent(v_APNandPCOs.extdProtocolConfigurationOptions)) {
      v_ExtdPcoFromUE := v_APNandPCOs.extdProtocolConfigurationOptions;
    }
    
    if (isvalue(v_PcoFromUE)) {
      v_IPv4AllocationViaNasFlag := f_CheckPCOforIPallocationViaNas(v_PcoFromUE);

    } else {
      v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_ExtdPcoFromUE);

    }
    
    if (pc_EPC_RACS and isvalue (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.sECURITY_MODE_COMPLETE.ueRadioCapId)) { // @sic R5-216280 sic@
      v_UERadioCapId := cs_UERadioCapId ('66'O, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.sECURITY_MODE_COMPLETE.ueRadioCapId.id);
    }
    
    v_EpsBearerId := f_EUTRA38_MultiPDN_InitialRegistration_Step14_15(p_CellId,
                                                                      NORMAL,
                                                                      v_AdditionalUpdateType,
                                                                      v_EPS_TI,
                                                                      v_IPv4AllocationViaNasFlag,
                                                                      v_PcoFromUE,
                                                                      v_APNandPCOs.accessPointName,
                                                                      cs_GprsTimer_v_deact,
                                                                      omit,
                                                                      omit,
                                                                      p_EquivalentPlmnList,
                                                                      0,
                                                                      omit,
                                                                      -,
                                                                      v_ExtdPcoFromUE,
                                                                      p_PduSessionInfoList,  // @sic R5s210149 sic@
                                                                      v_UERadioCapId); // @sic R5-216280 sic@
    
    // Step 10b7
    if (pc_noOf_PDNsSameConnection > 0 or FallbackSignalling) {  // @sic R5-206296 sic@
      /* In case of another PDN connectivity request this may be sent immediately by the UE
      => we cannot wait after the ATTACH_COMPLETE for any trigger from the IMS_PTC but need to do this when all PDNs are configured */
      v_WaitForUserplaneSignalling := noDelay;
    } else {
      select (f_EUTRA38_MultiPDN_GetPDNType(v_APNandPCOs.accessPointName)) { // @sic R5-206296 sic@
        case (IMS_DNN) {
          v_WaitForUserplaneSignalling := waitForIMS;
        }
        case (Internet_DNN) {
          v_WaitForUserplaneSignalling := dontWaitForIMS;
        }
      }
    }
    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId, v_WaitForUserplaneSignalling);
  }

